



CMS Pipelines Basics

Rick Barlow
Velocity Software

June 2026

Agenda

- Introduction
- What are Pipelines?
- PIPE syntax
- Pipeline stages
- Simple pipes
- Categories of stages
- Reference Information
- Acknowledgements
- Pipes in REXX
- REXX Interface sample programs
DUMPVARS
DUMPREXX
- Appendices
(lists of Pipeline stages)

Introduction

- This session is intended as an INTRODUCTION to using the CMS PIPE command as a part of programming an application under CMS using REXX.
- A working knowledge of coding in REXX is assumed.
- You should not expect to leave understanding everything there is to know about the CMS PIPE command. You should understand the basic concept of PIPEs.
- Significant documentation of the various PIPE stages is available through the VM HELP facility and through the authors PIPE AHELP stage.

What Are Pipelines?

- The Pipeline Concept

- A pipeline is simply a series of programs through which data flows, just as water flows through the sections of a water pipe. In a pipeline, a complex task is performed by processing data through several simple programs in an appropriate sequence.



- A word of caution:

- It is important to remember that, in most cases, all records placed into a pipeline must flow all the way through. This can dramatically affect the performance of a pipeline which reads a large file to select only a portion of the records.




What Are Pipelines?

- CMS Pipelines are made of programs called “Stages”
- Each stage:
 - Reads data from the pipeline
 - Processes (optionally transforms) it in some way
 - Writes the (transformed) data back to the pipeline
- Data is automatically presented as input to the next stage
- Individual stages are totally independent of one another
 - They do not need to know what comes before or after.
- They are also device independent
 - read from or write to any input or output device.

What Are Pipelines?

- 'Pipe-Think'

Pipe-Think is a term that was created to help conceptualize that it is important to think as a plumber when attempting to code pipelines. The flow of data through a pipeline does not stop at each stage. The first data into the Pipe passes to the next stage as soon as the stage has completed its function. The effect is that data can be processed in multiple stages at the same time.



PIPE Syntax

- The PIPE command format:

PIPE pipeline-specification (pipe options)

The argument to PIPE is a "pipeline specification". A pipeline specification is a string of characters listing the stages to be executed.

PIPE stage-1 | stage-2 | stage-3 | stage-4 | ... | stage-n

- The stage separator
 - Separates the stages in pipeline specification
 - Default is vertical bar (“|”) - x'4F'
 - PC keyboard default -> Shift-Backslash
 - Change with the STAGESEP or SEParator option on the PIPE command
 - If you want to use exclamation point (!) for a stage separator, a PIPE command might look like this:
PIPE (STAGESEP !) stage-1 ! stage-2 ! ... ! stage-n

PIPE Syntax

- The end character:
 - Used to concatenate multiple pipeline programs into a single pipeline specification
 - Necessary for stages which can accept multiple input streams or write multiple output streams
 - A frequently used end character is tilde (“~”) (Shift-back-quote on the keyboard)
 - Specify using the ENDchar or ESCape option on the PIPE command
 - PIPE (ENDCHAR ~) stage-1a | stage-1b ~ stage-2a | stage-2b

Pipeline Stages

- The stages which come with CMS Pipelines can be organized into three main categories
 - Device Drivers
 - interaction between the pipeline and the environment where it runs
 - The majority of the current Device Driver stages are listed in Appendix One.
 - Filters
 - provide the ability to modify the contents of the stream of data flowing through the pipeline by various types of selection, conversion and modification
 - The majority of the current Filter stages are listed in Appendix Two.
 - Other stages
 - Control
 - Gateways which provide for branching within the pipeline
 - Host Command Processors
 - Others miscellaneous stages
 - The majority of the current Other stages are listed in Appendix Three.

Pipeline Construction

- **Parts of a Pipeline**

- The correct term for input to and output from Pipeline Stages is a stream
- Most stages can accept one or more input streams
- Most stages can generate one or more output streams
- Some Device Driver stages only handle input or output
 - A stage that is input only must be the first stage in a specification.
 - A stage that is output only must be the last stage in a specification.
- Connections for multiple streams are coded in a Pipeline specification using labels
 - A label is a string of up to 8 alphanumeric characters followed by a colon
 - First occurrence is the label definition
 - Subsequent use is called a label reference

Let's get started – Device Driver Stages

- A simple pipe may contain only device drivers. Here is a traditional one to start with:


```
PIPE LITERAL Hello, World! | CONSOLE
Hello, World!
```

- Another simple pipe performs an echo operation:

```
PIPE CONSOLE | CONSOLE
```

This pipeline continues reading from the console and writing back until it reaches end-of-file, i.e., until it receives a null line as input.

```
Type first input
Type first input
Type second input
Type second input
<enter> to quit
```



Simple PIPEs

- All pipeline stages perform their designated function as well as passing all of the data through to subsequent stages.

- Consider this example:

PIPE CONSOLE | CONSOLE | CONSOLE

- What will happen?

```
pipe console | console | console
```

Test1

Test1

Test1

Test2

Test2

Test2



Simple PIPEs

- Simple pipelines using only device drivers could be executed to copy files from any valid device and/or file to any other device or file.
 - PIPE < PROFILE EXEC A | > PROFILE COPY A
 - PIPE < PROFILE EXEC A | CONSOLE
 - PIPE CONSOLE | FILEFAST NEW FILE A
 - 'PIPE FILEFAST' fn ft fm ,
 '| >' outfn outft outfm ,
 '| CONSOLE',
 '| PUNCH'

- Pipelines built only of device drivers do not really show the power of CMS Pipelines. There are dozens of built-in programs included with CMS Pipelines. Most of them are "filters"; programs that can be put into a pipeline to perform some transformation on the data flowing through the pipeline.
- A simple pipeline consisting of a couple of device drivers wrapped around a few filter stages can provide an instant enhancement to the CMS command set. Once you have some practice, you may find yourself typing lots of little "throwaway" pipes right on the command line.

Buffering

BUFFER
ELASTIC
SORT

Collect all records in a stage before passing any
Buffer only enough to prevent stall of following stages
Arrange records in ascending or descending order

Cut and Paste

SPLIT
STRIP

Split records into multiple records
Remove leading and/or trailing characters from records

Record Format

SCM

SPEC

Line up comments and complete unclosed comments in
REXX or C programs
Rearranges the contents of records

Selection Filters

DROP

Discards one or more records

PICK

Select records that satisfy selection criteria

FIND

Select records that begin with a specified text

LOCATE

Select records that contain a specified string of characters

NFIND

Select records that do not begin with a specified text

NLOCATE

Select records that do *not* contain a specified string

TAKE

Select one or more records from the beginning or end of the primary input stream

UNIQUE

Compare the contents of adjacent records and discards or retains the duplicate records

Other Filters

COUNT

Count bytes, blank-delimited character strings, records, or the length of the longest or shortest line

- Most of the filter programs are self-explanatory
- Often behave like the XEDIT subcommand of the same name
- Records that match the criteria in a filter stage are passed to the primary output stream – the next stage in the Pipeline
- If a label is coded on the filter stage, records that do not match the criteria are passed to the secondary output stream if it is connected

- This pipeline will locate a user-id within the response of the CP QUERY NAMES command:

```
PIPE CP QUERY NAMES | LOCATE /ZWEB/ | CONSOLE
ZWSSL03 - DSC , ZWSSL02 - DSC , ZWSSL01 - DSC , ZWEB05 - DSC
ZWEB04 - DSC , ZWEB03 - DSC , ZWEB02 - DSC , ZWEB01 - DSC
ZWEBLOG - DSC , ZWRITE - DSC , SSL00001 - DSC , ZVWS - DSC
```

- Notice that you get an entire line displayed. You could split the response at commas and remove leading blanks with:

```
PIPE CP QUERY NAMES | SPLIT AT , | STRIP | LOCATE /ZWEB/ | CONSOLE
ZWEB05 - DSC
ZWEB04 - DSC
ZWEB03 - DSC
ZWEB02 - DSC
ZWEB01 - DSC
ZWEBLOG - DSC
```

- Another filter much like LOCATE is FIND. These filters behave much like their counterparts in the XEDIT environment.
- The same two pipelines we just looked at with FIND in place of LOCATE:

```
PIPE CP QUERY NAMES | FIND ZWEB | CONSOLE
ZWEB04 - DSC , ZWEB03 - DSC , ZWEB02 - DSC , ZWEB01 - DSC
ZWEBLOG - DSC , ZWRITE - DSC , SSL00001 - DSC , ZVWS - DSC
PIPE CP QUERY NAMES | SPLIT AT , | FIND ZWEB | CONSOLE
ZWEB04 - DSC
ZWEBLOG - DSC
```

- We expected to find all occurrences of ZWEB.
What happened?
 - The FIND filter always looks for the data starting in column 1.

```
PIPE CP QUERY NAMES | SPLIT AT , | STRIP | FIND ZWEB | CONSOLE
ZWEB05 - DSC
ZWEB04 - DSC
ZWEB03 - DSC
ZWEB02 - DSC
ZWEB01 - DSC
ZWEBLOG - DSC
```

- The FIND filter also does not use a delimiter to determine the start and end of the search string. Therefore, it uses all of the characters until it finds a space or stage separator.
- If we only want to find the single user-id RSCS, we need to use an underscore to signify that a blank should follow RSCS. This is the same syntax as the XEDIT FIND subcommand.

```
PIPE CP QUERY NAMES | SPLIT AT , | STRIP | FIND RSCS_ | CONSOLE  
RSCS      - DSC
```

- There are filters to select all records that do **not** match the criteria
- If you intend to process both the records that do and do not meet the criteria, you can use either the positive or negative version of the selection stage with a label
- NLOCATE is the pipeline stage to exclude records that include a given string anywhere in the record.

```
PIPE CP QUERY NAMES | SPLIT AT , | NLOCATE /DSC/ | TAKE 5 | CONSOLE
JOE      -L0005
TYLER    -L0004
SAM      -L0003
JONATHAN -L0007
VSM      - TCPIP2
```

- NFIND is the pipeline stage to exclude records which begin with a specified string.

PIPE	CMS	QUERY	ACCESSED	NFIND Mode	CONSOLE
A	R/W	112	191	RRB191	
B	R/W	41	DIR	VSIPRV:RICKB.WORK.PIPE_CLASS	
C	R/W	1404	DIR	VSIPRV:RICKB.TOOLS	
L	R/O	1570	31A	SYSTLS	
P	R/O	117	319	SYS720	
S	R/O	693	190	MNT190	
Y/S	R/O	1116	19E	MNT19E	

- Another useful filter is COUNT. It can return various characteristics about the data in your pipeline including the number of lines or words.
- If you want to know how many records are in a file, you could type:

```
PIPE CP QUERY NAMES | COUNT LINES | CONSOLE  
12
```

- If you want to know how many words are in a file, you could type:

```
PIPE CP QUERY NAMES | COUNT WORDS | CONSOLE  
163
```

- You could even determine the number of unique words with:

```
PIPE CP QUERY NAMES | SPLIT | SORT UNIQUE | COUNT WORDS | CONSOLE  
48
```

- **TAKE [FIRST|LAST] [n]**
 - Select 1 or more records from one end of the stream of data
 - FIRST - select from the beginning
 - LAST – select from the end
 - [n] is the optional number of records – default is 1
- **DROP [FIRST|LAST] [n]**
 - Omit 1 or more records from one end of the stream of data
 - FIRST - select from the beginning
 - LAST – select from the end
 - [n] is the optional number of records – default is 1

- **SORT**
 - Re-order the records in the stream
 - This stage will cause all records to be delayed (buffered)
 - Sort by word(s) and/or column(s)
 - Combine with other stages for specific results
- **UNIQUE**
 - Use to cause a single copy of a given record / word / columns to pass to the primary output stream
 - FIRST – keep the first record with the selected criteria
 - LAST – keep the last record; this is the default

- Using device drivers and filters can be very useful for writing “throw away” Pipelines at the CMS command line.
- Filters become most useful when used in combination.
- Suppose you wanted to see the last disk that was accessed in your virtual machine. You could use the SORT and TAKE stages like this:

```
PIPE CMS QUERY ACCESSED | NLOCATE 1.4 /Mode/ | SORT 1.1 D | TAKE | CONSOLE  
Y/S      R/O          1232  19E    MNT19E
```

- Even more simply

```
PIPE CMS QUERY ACCESSED | NFIND Mode | SORT W1 D | TAKE | CONSOLE  
Y/S      R/O          1232  19E    MNT19E
```

Other Stages

Control Stages

APPEND

Write primary input stream records to the primary output stream followed by records from a specified device driver

Gateways (pipe fittings)

FANOUT

Copy primary input stream records to multiple output streams

FANIN

Combine multiple input streams into a single stream in a specified order

FANINANY

Combine multiple input streams into a single stream. FANINANY reads an input record from any input stream that has a record available.

GATHER

Read records from all connected streams in sequential order or other specified order

Host Command Processors

CMS

Issue CMS commands with full command resolution

COMMAND

Issue CMS commands as if they were invoked using **Address 'COMMAND'** from REXX/VM

CP

Issue CP commands

Host Command Stages

- Issue host commands and routes the responses into the pipeline.
 - CP issues CP commands
 - COMMAND issues CMS commands using program call
 - Parsing behaves like REXX Address 'COMMAND'
 - Uses extended parameter list
 - CMS issues CMS commands with full command resolution
 - Parsing behaves like the CMS command line or REXX Address 'CMS'
 - Uses CMS subcommand environment
- Each of these stages issues its argument string as a command and then reads the input stream and issues each record as a commands. The command responses are captured and each line becomes a record in the pipeline.

```
PIPE LITERAL QUERY TIME | CP QUERY USERID | CONSOLE
RICKB      AT VSIVM1
TIME IS 21:55:24 EST WEDNESDAY 2024-01-10
CONNECT= hh:mm:ss VIRTCPU= mmm:ss.hh TOTCPU= mmm:ss.hh
```

- The return code of each command is written to the secondary output stream

Host Command Stages

- A simple pipeline to talk to CMS might look like this:

```
PIPE LITERAL QUERY ACCESSED | CMS IDENTIFY | CONSOLE
```

If you place it in an EXEC, it might look like this:

```
/* Who am I and what is my environment like */  
'PIPE(NAME CLASS)',  
  '| LITERAL QUERY ACCESSED',  
  '| CMS IDENTIFY',  
  '| CONSOLE'
```

```
RICKB      AT VSIVM1    VIA RSCS      2026-01-10 15:07:25 EST      WEDNESDAY  
Mode  Stat      Files  Vdev  Label/Directory  
A      R/W       2299  191   RB191B  
S      R/O        709  190   MNT190  
Y/S    R/O      1232  19E   MNT19E
```

More Built-in Programs: SPEC

- One of the most complex and least obvious stages
 - SPEC stage design influenced by the IBM 407 Accounting Machine
- Simple form: selects one or more pieces of an input record and puts them into the output record
 - SPEC input-location1 output-location1 ...
 - Specify as many input/output pairs as you need
- Basic syntax is like the SPEC option of the CMS COPYFILE
- Includes many more features

More Built-in Programs: SPEC

- Input location options

- Location

- start-end

e.g. 1-* or 10-14

- start.length

e.g. 10.5

- wordrange or wnumberrange

- select blank-delimited word(s)

- examples

WORD 2.5 Word 8 W -2;-1

- literal

- a delimited string

e.g. /Some Thing/

- RECNO

- the current records number in the stream

More Built-in Programs: SPEC

- Output location options
All output location options may include a length
 - Explicit location
 - starting column number e.g. 10
 - start-end e.g. 10-14
 - start.length e.g. 10.5
 - Relative location
 - **N**ext column of the record e.g. NEXT or N.8
 - **N**ext**W**ord – leaves one space e.g. NEXTWORD or NW.4

More Built-in Programs: SPEC

- More complex features include
 - numeric calculations
 - field summarization
 - report generation
- SPEC documentation
 - Chapter 16 of the [CMS/TSO Pipelines: Author's Edition](#)
 - PIPE AHELP SPECTUT
 - CMS/TSO Pipelines Runtime Library Distribution web site
 - <http://vm.marist.edu/~pipeline/ahelp/spectuto.html>

More Built-in Programs: SPEC

- Examples

```
PIPE LITERAL Hello, there.|SPEC W1 1 W-1;-1 15|CONSOLE  
Hello,           there.
```

```
PIPE LITERAL Hello, there.|SPEC 1-5 1.10 6 NW 8-* 20|CONS  
Hello           ,           there.
```

```
|...+....1....+....2....+....3....+....4....+....5
```

More Built-in Programs: SPEC

- **Conversion routine**
 - Add between input and output location
 - Similar to REXX conversion function
 - C2B, C2D, C2X, D2C, X2C...
 - Applied to input field before moving to output field

- **Examples**

```
PIPE LITERAL Hello|SPEC 1-* C2X 1|CONSOLE
```

```
C885939396
```

```
PIPE LITERAL C885939396|SPEC 1-* X2C 1|CONSOLE
```

```
Hello
```

More Built-in Programs: SPEC

- **Placement option**
 - Add after output location
 - Cause alignment in the output location
 - "LEFT", "CENTER" or "RIGHT"

- Example

```

PIPE LITERAL Record Number
    | SPEC W1 1 W2 NW.10 RIGHT RECNO 46.4 RIGHT
    | CONSOLE

```

Record	Number	1
--------	--------	---

$$| \dots + \dots 1 \dots + \dots 2 \dots + \dots 3 \dots + \dots 4 \dots + \dots 5$$

More Built-in Programs: SPEC

- Suppose you want to take the list of disks we created before and show only the virtual address and access mode. You could use the following pipeline:

```
PIPE COMMAND QUERY ACCESSED | SPEC W1 1 W4 NW | CONSOLE
```

```
Mode Vdev
```

```
A 191
```

```
S 190
```

```
X DIR
```

```
Y/S 19E
```

- Perhaps you would like to get the last 2 words of a record. To do that, you use the 'wnumberrange' format like this:

```
PIPE COMMAND QUERY ACCESSED | SPEC WORD -2;-1 1 | CONSOLE
```

```
Vdev Label/Directory
```

```
191 RB191B
```

```
190 MNT190
```

```
DIR VSIPRV:RICKB.STUFF
```

```
19E MNT19E
```

SPEC Examples

- Find out what disks and/or directories are accessed

```
PIPE COMMAND QUERY ACCESSED | SPECS W1 1 W4.2 NW | CONSOLE
Mode Vdev Label/Directory
A 191 RB191B
S 190 MNT190
X DIR VMPOOL:RICKB.EXTRA_SPACE
Y/S 19E MNT19E
```

- Modify the Pipeline to show just the access mode, VDEV or DIR and the Label/Directory

```
PIPE COMMAND QUERY ACCESSED | SPECS W1 1 W4 NW W5 NW | CONSOLE
Mode Vdev Label/Directory
A 191 RB191B
X DIR VMPOOL:RICKB.EXTRA_SPACE
S 190 MNT190
Y/S 19E MNT19E
```

SPEC Examples

- Modify the Pipeline to line up the columns

```
PIPE COMMAND QUERY ACCESSED | SPECS W1 1.4 W4 NW.4 W5 NW | CONSOLE
Mode Vdev Label/Directory
A 191 RB191B
X DIR VMPOOL:RICKB.EXTRA_SPACE
S 190 MNT190
Y/S 19E MNT19E
```

- Modify the Pipeline to number the lines of output

```
PIPE COMMAND QUERY ACCESSED | SPECS RECNO 1.2 RIGHT W1 NW.4 W4 NW.4 W5 NW |
CONSOLE
1 Mode Vdev Label/Directory
2 A 191 RB191B
3 S 190 MNT190
4 X DIR VMPOOL:RICKB.VM.STUFF
5 Y/S 19E MNT19E
```

Host Command - Build a Host Command

- A simple REXX EXEC to talk to CP might look like this:

```
PIPE CP Q USERID|CONSOLE|SPEC /Q/ 1 W1 NW|CP|CONSOLE
```

```
/* Who am I and where am I logged on */
```

```
'PIPE(NAME SLIDE41:03) ',  
  '| CP QUERY USERID',  
  '| CONSOLE',  
  '| SPEC /QUERY/ 1 Word 1 NextWord',  
  '| CONSOLE',  
  '| CP',  
  '| CONSOLE'
```

- The resulting output would resemble this:

```
RICKB      AT VSIVM1
```

This is the result of the CP stage

```
QUERY RICKB
```


This is the command generated by the SPEC stage

```
RICKB      -L00E4
```

This is the result of the second CP stage

Gateway Stages

- The majority of the Gateway stages are similar to the fittings that you might include in a water pipe. You might look at them like a 'T'. The most common are the FANOUT, FANIN, FANINANY and GATHER stages.
- When a record in the stream is rejected by a selection filter, the record is placed on the secondary stream of that stage. Labels are used to process the records in an additional pipeline.



Gateway Stages

- For example, we could assemble a pipeline where we send the CMS Query disk output into two streams like this:

```
/* */
Address 'COMMAND'
'PIPE(ENDCHAR ?)',
  '| CMS QUERY ACCESSED',
  '| DROP',
  '| O: FANOUT',
  '| SPEC /1/ 1 1-69 NW',
  '| LITERAL This stuff is in the first stream:',
  '| CONSOLE',
'? O:',
  '| SPEC /2/ 1 1-69 NW',
  '| LITERAL This stuff is in the second stream:',
  '| CONSOLE'
```

Gateway Stages


- Result:

This stuff is in the first stream:

This stuff is in the second stream:

```
1 A      R/W      112  DIR   SFSVM1:userid.
2 A      R/W      112  DIR   SFSVM1:userid.
1 B      R/W     1404  DIR   VSIPRV:userid.SHAREDTOOLS
2 B      R/W     1404  DIR   VSIPRV:userid.SHAREDTOOLS
1 S      R/O      693  190   MNT190
2 S      R/O      693  190   MNT190
1 Y/S    R/O     1116  19E   MNT19E
2 Y/S    R/O     1116  19E   MNT19E
1 Z      R/W       41  DIR   VSIPRV:RICKB.WORK.PIPE_CLASS
2 Z      R/W       41  DIR   VSIPRV:RICKB.WORK.PIPE_CLASS
```

- What is wrong?



Gateway Stages

- To get the output in the order we expect, one thing we can do is to delay the data in the second pipeline:

```
/* */
'PIPE (ENDCHAR ?)',
'| CMS QUERY ACCESSED',
'| DROP',
'| O: FANOUT',
'| SPEC /1/ 1 1-69 NW',
'| LITERAL This stuff is in the first stream:',
'| CONSOLE',
'? O:',
'| SPEC /2/ 1 1-69 NW',
'| LITERAL This stuff is in the second stream:',
'| BUFFER',
'| CONSOLE'
```

Gateway Stages


- Result:

This stuff is in the first stream:

```
1 A      R/W      112  DIR   SFSVM1:userid.  
1 B      R/W     1404  DIR   VSIPRV:userid.SHAREDTOOLS  
1 S      R/O      693  190   MNT190  
1 Y/S    R/O     1116  19E   MNT19E  
1 Z      R/O      41   DIR   VSIPRV:RICKB.WORK.PIPE_CLASS
```

This stuff is in the second stream:


```
2 A      R/W      112  DIR   SFSVM1:userid.  
2 B      R/W     1404  DIR   VSIPRV:userid.SHAREDTOOLS  
2 S      R/O      693  190   MNT190  
2 Y/S    R/O     1116  19E   MNT19E  
2 Z      R/O      41   DIR   VSIPRV:RICKB.WORK.PIPE_CLASS
```



Gateway Stages

- To eliminate two CONSOLE stages, we could use a FANIN stage.

```
/* */
Address 'COMMAND'
'PIPE (NAME SLIDE48:001 ENDCHAR ?)',
  '| CMS QUERY ACCESSED',
  '| DROP',
  '| O: FANOUT',
  '| SPEC /1/ 1 1-69 NW',
  '| LITERAL This stuff is in the first stream:',
  '| I: FANIN',
  '| CONSOLE',
'? O:',
  '| SPEC /2/ 1 1-69 NW',
  '| LITERAL This stuff is in the second stream:',
  '| ELASTIC',
  '| I:'
```



Gateway Stages

Another way that the data can be split into multiple streams is through selection filters. Data that is selected goes to the primary output stream. Unselected data passes to the secondary output stream. For example:


```
/* */
Address 'COMMAND'
'PIPE (NAME SLIDE49:001 ENDCHAR ?)',
'| CMS QUERY ACCESSED',
'| DROP',
'| L: LOCATE ,R/W,',
'| LITERAL The Read/Write disks are',
'| F: FANIN',
'| CONSOLE',
'? L:',
'| LITERAL The Read Only disks are',
'| ELASTIC',
'| F:'
```

The Read/Write disks are

A	R/W	60	DIR	SFSVM1:userid.
B	R/W	0	DIR	VSIPRV:userid.SHAREDTOOLS

The Read Only disks

S	R/O	693	190	MNT190
Y/S	R/O	1116	19E	MNT19E
Z	R/O	41	DIR	VSIPRV:RICKB.WORK.PIPE_CLASS



- “Throw away” – Pipelines used at the command line
 - The bigger your system(s), the more useful
 - Filter responses from CP commands
 - Query NAMES – select the ones you want to see
 - Use selections from CP commands to build and run more commands
 - Query devices – select the ones you want; generate more CP commands; execute; filter the responses
- Pipelines in REXX for repeated use

More Examples

- Messaging CP command output

PIPE	CP	QUERY	VIRTUAL	DASD	CONSOLE						
DASD	0190	3390	VM1004	R/O	107	CYL	ON	DASD	320C	SUBCHANNEL = 000C	
DASD	0191	3390	VM1004	R/W	50	CYL	ON	DASD	320C	SUBCHANNEL = 0004	
DASD	019E	3390	VM1004	R/O	200	CYL	ON	DASD	320C	SUBCHANNEL = 000E	

More Examples

- Use the output from the CP command to build commands to issue QUERY MDISK DETAILS for each device

```
PIPE CP QUERY VIRTUAL DASD | CONSOLE | SPECS /QUERY MDISK/ 1 W2 NW /DETAILS/ NW | CP | CONSOLE
DASD 0190 3390 VM1004 R/O          107 CYL ON DASD 320C SUBCHANNEL = 000C
TargetID Tdev OwnerID  Odev Minidisk  DEVNO Duplex
RICKB    0190 MAINT    0190 Regular   No    Primary
DASD 0191 3390 VM1004 R/W          50 CYL ON DASD 320C SUBCHANNEL = 0004
TargetID Tdev OwnerID  Odev Minidisk  DEVNO Duplex
RICKB    0191 RICKB    0191 Regular   No    Primary
DASD 019E 3390 VM1004 R/O          200 CYL ON DASD 320C SUBCHANNEL = 000E
TargetID Tdev OwnerID  Odev Minidisk  DEVNO Duplex
RICKB    019E MAINT    019E Regular   No    Primary
```

More Examples

- Improve readability and Insert blank lines

```
PIPE CP QUERY VIRTUAL DASD | CHANGE // ` / | SPLIT AT ` | CONSOLE | LOCATE W1 |  
SPECS /QUERY MDISK/ 1 W2 NW /DETAILS/ NW | CP | CONSOLE
```

```
DASD 0190 3390 VM1004 R/O          107 CYL ON DASD 320C SUBCHANNEL = 000C  
TargetID Tdev OwnerID  Odev Minidisk  DEVNO Duplex  
RICKB    0190 MAINT    0190 Regular   No    Primary
```

```
DASD 0191 3390 VM1004 R/W          50 CYL ON DASD 320C SUBCHANNEL = 0004  
TargetID Tdev OwnerID  Odev Minidisk  DEVNO Duplex  
RICKB    0191 RICKB    0191 Regular   No    Primary
```

```
DASD 019E 3390 VM1004 R/O          200 CYL ON DASD 320C SUBCHANNEL = 000E  
TargetID Tdev OwnerID  Odev Minidisk  DEVNO Duplex  
RICKB    019E MAINT    019E Regular   No    Primary
```

More Examples

- Issue CP command and filter the results; show the first 10 lines of output
Note that this requires a user-id with Class B privilege.

```
PIPE CP QUERY DASD ALL | TAKE FIRST 10 | CONSOLE
```

DASD	2000	CP	SYSTEM	64SRCP	1	
DASD	2004	CP	SYSTEM	VM1001	0	SHARED
DASD	2005	CP	SYSTEM	VM1004	0	SHARED
DASD	2023	CP	OWNED	VMCOM1	12	
DASD	2024	CP	OWNED	M01P01	0	
DASD	2059	CP	OWNED	V3COM2	0	
DASD	206F	CP	SYSTEM	RBMOD9	1	SHARED
DASD	20F0	CP	SYSTEM		0	
DASD	20F1	CP	SYSTEM		0	
DASD	20F2	CP	SYSTEM		0	

More Examples

- Issue CP command and filter the results; show the first 10 and last 10 lines of output

```
PIPE (endchar ~)| CP QUERY DASD ALL| T:TAKE FIRST 10 | F:FANIN| CONSOLE ~ T:| TAKE LAST 10 | ELASTIC | F:
```

```
DASD 2000 CP SYSTEM 64SRCP 1
DASD 2004 CP SYSTEM VM1001 0 SHARED
DASD 2005 CP SYSTEM VM1004 0 SHARED
DASD 2023 CP OWNED VMCOM1 12
DASD 2024 CP OWNED M01P01 0
DASD 2059 CP OWNED V3COM2 0
DASD 206F CP SYSTEM RBMOD9 1 SHARED
DASD 20F0 CP SYSTEM 0
DASD 20F1 CP SYSTEM 0
DASD 20F2 CP SYSTEM 0
DASD 23BB BU23BB , DASD 23BC BU23BC , DASD 23BD BU23BD , DASD 23BE BU23BE
DASD 23BF BU23BF , DASD 23C0 BU23C0 , DASD 23C1 BU23C1 , DASD 23C2 ZZ23C2
DASD 23C3 ZZ23C3 , DASD 23C4 ZZ23C4 , DASD 23C5 ZZ23C5 , DASD 23C6 ZZ23C6
DASD 23C7 ZZ23C7 , DASD 23C8 ZZ23C8 , DASD 23C9 ZZ23C9 , DASD 23CA ZZ23CA
DASD 23CB ZZ23CB , DASD 23CC ZZ23CC , DASD 23CD ZZ23CD , DASD 23CE ZZ23CE
DASD 23CF ZZ23CF , DASD 2800 RANW00 , DASD 2801 RANW01 , DASD 2802 RANW02
DASD 2803 ZZ2803 , DASD 2804 OPSH00 , DASD 2805 OPSH01 , DASD 2806 OPSH02
DASD 2807 OPSH03 , DASD 2808 OPSH04 , DASD 2809 OPSH05 , DASD 280A OPSH06
DASD 280B OPSH07
DASD 214A OFFLINE
```

More Examples

- It is important to know the data you are processing
- It often requires some experimentation to get the results to look as you intend
 - Show the first and last 10 after removing OFFLINE

```
'PIPE (NAME SLIDE55:003 ENDCHAR ~)',
'| CP QUERY DASD ALL',
'| NLOCATE /OFFLINE/',
'| T:TAKE FIRST 10',
'| F:FANIN',
'| CONSOLE',
'~ T:',
'| TAKE LAST 10',
'| ELASTIC',
'| F:'
```

DASD	2000	CP	SYSTEM	64SRCP	1	
DASD	2004	CP	SYSTEM	VM1001	0	SHARED
DASD	2005	CP	SYSTEM	VM1004	0	SHARED
DASD	2023	CP	OWNED	VMCOM1	12	
DASD	2024	CP	OWNED	M01P01	0	
DASD	2059	CP	OWNED	V3COM2	0	
DASD	206F	CP	SYSTEM	RBMOD9	1	SHARED
DASD	20F0	CP	SYSTEM		0	
DASD	20F1	CP	SYSTEM		0	
DASD	20F2	CP	SYSTEM		0	
DASD	23B7	24AU02		DASD	23B8	ZZ23B8 , DASD 23B9 ZZ23B9 , DASD 23BA ZZ23BA
DASD	23BB	BU23BB		DASD	23BC	BU23BC , DASD 23BD BU23BD , DASD 23BE BU23BE
DASD	23BF	BU23BF		DASD	23C0	BU23C0 , DASD 23C1 BU23C1 , DASD 23C2 ZZ23C2
DASD	23C3	ZZ23C3		DASD	23C4	ZZ23C4 , DASD 23C5 ZZ23C5 , DASD 23C6 ZZ23C6
DASD	23C7	ZZ23C7		DASD	23C8	ZZ23C8 , DASD 23C9 ZZ23C9 , DASD 23CA ZZ23CA
DASD	23CB	ZZ23CB		DASD	23CC	ZZ23CC , DASD 23CD ZZ23CD , DASD 23CE ZZ23CE
DASD	23CF	ZZ23CF		DASD	2800	RANW00 , DASD 2801 RANW01 , DASD 2802 RANW02
DASD	2803	ZZ2803		DASD	2804	OPSH00 , DASD 2805 OPSH01 , DASD 2806 OPSH02
DASD	2807	OPSH03		DASD	2808	OPSH04 , DASD 2809 OPSH05 , DASD 280A OPSH06
DASD	280B	OPSH07				

More Examples

- It is important to know the data you are processing
- It often requires some experimentation to get the results to look as you intend
 - Show the first and last 10 after removing OFFLINE and devices with no label

```
'PIPE (NAME SLIDE56:004 ENDCHAR ~)' ,
'| CP QUERY DASD ALL' ,
'| NLOCATE /OFFLINE/' ,
'| LOCATE W6' ,
'| T:TAKE FIRST 10' ,
'| F:FANIN' ,
'| CONSOLE' ,
'~ T:' ,
'| TAKE LAST 10' ,
'| ELASTIC' ,
'| F:'
```

```
DASD 2000 CP SYSTEM 64SRCP 1
DASD 2004 CP SYSTEM VM1001 0 SHARED
DASD 2005 CP SYSTEM VM1004 0 SHARED
DASD 2023 CP OWNED VMCOM1 12
DASD 2024 CP OWNED M01P01 0
DASD 2059 CP OWNED V3COM2 0
DASD 206F CP SYSTEM RBMOD9 1 SHARED
DASD 2100 CP SYSTEM 64SRCB 1
DASD 2104 CP SYSTEM VM1002 0 SHARED
DASD 2123 CP SYSTEM C373R1 12 SHARED
DASD 23B3 24ADB1 , DASD 23B4 24AU03 , DASD 23B5 LOGR03 , DASD 23B6 24APG1
DASD 23B7 24AU02 , DASD 23B8 ZZ23B8 , DASD 23B9 ZZ23B9 , DASD 23BA ZZ23BA
DASD 23BB BU23BB , DASD 23BC BU23BC , DASD 23BD BU23BD , DASD 23BE BU23BE
DASD 23BF BU23BF , DASD 23C0 BU23C0 , DASD 23C1 BU23C1 , DASD 23C2 ZZ23C2
DASD 23C3 ZZ23C3 , DASD 23C4 ZZ23C4 , DASD 23C5 ZZ23C5 , DASD 23C6 ZZ23C6
DASD 23C7 ZZ23C7 , DASD 23C8 ZZ23C8 , DASD 23C9 ZZ23C9 , DASD 23CA ZZ23CA
DASD 23CB ZZ23CB , DASD 23CC ZZ23CC , DASD 23CD ZZ23CD , DASD 23CE ZZ23CE
DASD 23CF ZZ23CF , DASD 2800 RANW00 , DASD 2801 RANW01 , DASD 2802 RANW02
DASD 2803 ZZ2803 , DASD 2804 OPSH00 , DASD 2805 OPSH01 , DASD 2806 OPSH02
DASD 2807 OPSH03 , DASD 2808 OPSH04 , DASD 2809 OPSH05 , DASD 280A OPSH06
```


More Examples

- Build a list of ZVWS user-ids
 - Here is one way to list all of the ZVWS user-ids

```
PIPE CP Q NAMES | SPLIT AT , | STRIP | FIND ZW | SPEC W1 1.9 | JOIN 4 | CONS
ZWSSL05 ZWSSL04 ZWSSL03 ZWSSL02 ZWSSL01
ZWAPI03 ZWAPI02 ZWAPI01 ZWEB05 ZWEB04
ZWEB03 ZWEB02 ZWEB01 ZWEBLOG ZWRITE
```

More Complex

- LOOKUP – Find records in a reference
- Both Master and Detail records need to be sorted the same



More Complex

```
/* MYLOOKUP EXEC */
'PIPE (NAME LOOKUP:001 ENDCHAR ?)',
'| < DETAIL RECORDS',          /* read detail records */
'| L: LOOKUP',                 /* find matches */
'| > MATCHING RECORDS A',      /* write matching masters and details */
'? < MASTER RECORDS',         /* start of second pipeline */
',                             /* read master records */
'| L:',                        /* define secondary input & output stream for LOOKUP */
'| > UNREF DETAILS A',         /* write details without masters */
'? L:',                        /* start of third pipeline */
',                             /* define tertiary output for LOOKUP */
'| > UNREF MASTERS A'          /* write masters without details */
Exit rc
```

- **HELP PIPE MENU**
- **PIPE AHELP MENU**
- SC24-6252.....CMS Pipelines User's Guide and Reference
- CMS/TSO Pipelines Runtime Library Distribution web site
<http://vm.marist.edu/~pipeline>

PIPE MENU
(c) Copyright IBM Corporation 1992, 2024

Menu Help Information

line 1 of 40

Help for CMS Pipelines stages and subcommands

To view a Help panel, move the cursor to any character of the name
and press the ENTER key or the PF1 key.
An asterisk (*) preceding the name indicates a MENU panel.
A colon (:) preceding the name indicates a TASK panel.

<	ASMCONT	COMMAND	ENBASE64	FULLSCR	IP2SOCKA	NLOCATE	PREFACE	SFSUPDate	STRFIND	TOKENIZE	VCHAR
<MDSK	ASMEXPAND	COMMIT	EDFBACK	FULLSCRQ	ISPF	NOCOMMIT	PREPEND	SHORT	STRFRLABe	TOLABe	VERIFY
<OE	ASMFIND	CONFIGURE	EDFREPOR	FULLSCRs	ISSUEMSG	NOEDFBACK	PRINTMC	SNAKE	STRFROMLa	TOTARGe	VMC
<SFS	ASMINFIND	CONFIGVA	ESCAPE	GATE	JEREMY	NOT	PUNCH	SOCKA2IP	STRIP	TRACKBLoc	VMCDATA
<SFSSLOW	ASMXPAND	CONSOLE	FANIN	GATHER	JOIN	NOTEOFBAC	QPDECODE	SOCKA2IP	STRITe	TRACKDEBl	VMCFCLIE
>	ASMXPND	COPY	FANINANY	GETfiles	JOINCNT	NOTFIND	QPENCODE	SPACE	STRLOCate	TRACKEXPa	VMCFDATA
>>	BEAT	COUNT	FANINTWO	GETRANGE	JUKTAPOSE	NOTINSIDE	QSAH	SPEC	STRNFIND	TRACKREAD	VMCFLISTe
>>MDSK	BEGOUTPU	CP	FANOUT	GREG2SEC	LDRTBLS	NOTLOCate	Query	SPECREFE	STRNLOCat	TRACKSQUI	VMCLIENT
>>OE	BETWEEN	CRC	FANOUTWO	HELP	LISTPDS	NUEXT	RANDOM	SPECTUTO	STRNDTLc	TRACKVERi	VMCLISTEN
>>SFS	BFS	C14TQ38	FBAREAD	HFS	LITERAL	NUL	READER	SPILL	STRTOLABe	TRACKWRI	WAITDEV
>>SFSSLOW	BFSDIRECT	DAM	FBWRITE	HFSDIRECT	Locate	OPTCDJ	READTO	SPLIT	STRUCTRE	TRACKXPan	WARP
>MDSK	BFSEXecut	DATECNVT	FBLOCK	HFSEXecut	LOOKUP	OUTPUT	RESOLVE	SQL	STRUCTure	TRANSLate	WARPLIST
>OE	BFSQuery	DATECONVe	FILEBACK	HFSQuery	MACLIB	OUTSIDE	RETAB	SQLCODES	STRWHILEl	TRANSLite	WHILELAB
>SFS	BFSREPlac	DEAL	FILEDESCr	HFSREPlac	MAPMDISK	OUTSTORE	REVERSE	SQLSELEC	STSI	TRFREAD	WILDCARD
ABBREV	BFSSTATE	DEBLOCK	FILEfast	HFSSTATE	MAXSTREA	OVERlay	REXX	STACK	SUBCOM	TRUNCate	XAB
ACIGROUP	BFSXecute	DELAY	FILERAND	HFSXecute	MCTOASA	OVERSTR	REXXCMD	STAGENUM	SUBSTRing	UDP	XEDIT
ADDPipe	BLOCK	DEVINFO	FILESLOW	HLASM	MDISKBLK	PACK	REXXVARS	STARMDN	SUSPEND	UNIQUE	XEDITMSG
ADDRDW	BROWSE	DFSORT	FILETOKE	HLASMERR	MDSKBACK	PAD	RUNPIPE	STARMSG	SYNChroni	UNPACK	XLATE
ADDRSPACE	BRW	DIAGE4	FILEUPDA	HOLE	MDSKBLK	PARCEL	SCANNER	STARSYS	SYNTAX	UNTAB	XMSG
ADDRSTREA	BUFFER	DIGEST	FILLUP	HOSTBYADD	MDSKFAST	PAUSE	SCANRANG	STATE	TAKE	UPDATE	XPNDHI
ADDRSPACE	BUILDSCR	DISKBACK	FILTERPac	HOSTBYNam	MDSKRAND	PDSdirect	SCANSTRI	STATEW	TAPE	URLDEBLc	XRANGE
AFTFST	CALLPIPE	DISKfast	FIND	HOSTID	MDSKSLOW	PEEKTO	SCM	STEM	TCPCSUM	URLDECODE	ZONE
AGGRc	CASEI	DISKID	FITTING	HOSTNAME	MDSKUPDA	PICK	SEC2GREG	STFLe	TCPCLIEnt	URD	3WAY
AHELP	CHANGE	DISKRAND	FLTPACKag	HTTPSPli	MEMBERS	PIPCMD	SELECT	STGSELEC	TCPDATA	UTF	3270BFRA
ALL	CHOP	DISKSLow	FMFST	IEBCOPY	MERGE	PIPDUMP	SETRC	STORAGE	TCPLISTEN	VAR	3270ENC
ALSERV	CKDDEBLc	DISKUPDA	FRLABe	IF	MESSAGE	PIPESTOP	SEVER	STRASMFIn	TERMinal	VARDROP	3277BFRA
APLDECode	CMD	DROP	FROMLABe	IMMCM	MULTVERS	PIPEVENT	SFSBACK	STRASMNFi	THREEWAY	VARFETCh	3277ENC
APLENCode	CMS	DUPlicate	FROMTARGe	INSERT	NFIND	POLISH	SFSDIRECT	STREAMNU	TIMEstamp	VARLOAD	64DECODE
APPEND	COLLATE	ELASTIC	FRTARGET	INSIDE	NINSIDE	PREDSELe	SFSRANDOM	STREAMST	TOKENISE	VARSET	64ENCODE
ASATOMC	COMBINE	EMSG	FTP	INSTORE							

*** End of File ***

PF1= Help 2= Top 3= Quit 4= Return 5= Clocate 6= ?
PF7= Backward 8= Forward 9= PFkeys 10= 11= 12= Cursor

====>

PIPEs in REXX

- If you want to include a PIPE command in a REXX EXEC, keep in mind that the entire command is a string, only portions of which should have variables substituted. For example, the last PIPE would look like this in a REXX EXEC:

```
'PIPE <' fn ft fm '|' >' outfn outft outfm '|' CONSOLE | PUNCH'
```

You should quote the parts that are not variable while allowing REXX to substitute the correct values for the variable fields.

- A PIPELINE written as a single continuous string is known as **linear** format.
- FMTP XEDIT is a macro to convert a **linear** Pipeline specification to **portrait** mode

```
'PIPE (name AN.EXEC:5) ',  
'| <' fn ft fm ,  
'| >' outfn outft outfm ,  
'| CONSOLE '  
'| PUNCH'
```

PIPEs in REXX – XEDIT Macros

- XEDIT macros have a maximum string length of 255 characters. Therefore, when including complex Pipelines in XEDIT macros it may be necessary to include Address 'COMMAND' on the PIPE command to avoid truncating the pipeline specification.
- Putting literals in quotes is important so that later coding does not change the program by using a word that you used as a literal for a variable.

REXX Coding Sideline

- Choose a coding style and stick to it
 - Improves readability when coding is consistent
 - Understand what is a command, literal, etc
- Things to consider
 - Case: all upper, all lower, mixed - when?
 - Indentations
 - Comments
 - Spaces around operators (= + - |)

REXX Coding Sideline

- Recommendations
 - REXX commands Capitalized
 - Pipelines
 - Stages Upper case
 - Separator to the left
 - Literals in single quotes in desired case
 - Variables all lower case
 - Indentations
 - 3 characters
 - End is outdented to align with the start of the group it ends

DUMPVARS EXEC

```
/*%I:Dump variable(s) to the screen (sub routine)                                :I%*/
Address 'COMMAND'
Parse Arg variable_names '(' holdopt .
Do var_index = 1 to Words(variable_names)
    variable_name = Word(variable_names,var_index)
    'PIPE VAR' variable_name '1 | VAR' variable_name
    Say variable_name '=' Value(variable_name)
End
If holdopt = 'HOLD' Then Parse External
Return

/* test DUMPVARS*/
Address 'COMMAND'
x='Hi!'
y='Y'all'
Call DumpVars('X Y')

x = Hi!
y = Y'all
```

DUMPREXX EXEC

```
/* */
Address 'COMMAND'
Arg opt .
'PIPE CP QUERY SET |',
  'TAKE 1 |',
  'CHANGE /, / / / |',
  'SPECS W 6 1 |',
  'VAR EMSGSET'
'CP SET EMSG OFF'
'ERASE DUMPREXX LISTING A'
rc = 0
Do i = 1 Until rc \= 0
  'PIPE (end ? name DUMPREXX) |',
    'REXXVARS' i '|',
    'VARS: TAKE 1 |',
    'SPECS "Command:" 1 W 2-* NW |',
    'APP: FANIN |',
    'LITERAL --> Level' i-1 'environment <--|',
    '>> DUMPREXX LISTING A',
    '? VARS: |',
    'BUFFER |',
    'CHANGE 1.2 /n / // |',
    'CHANGE 1.2 /v / /=/ |',
    'JOIN 1 |',
    'APP:'
  End
'CP SET EMSG' emsgset
If opt = '' Then 'XEDIT DUMPREXX LISTING A'
```

```
Command: CMS SUBROUTINE DUMPREXX EXEC L1 DUMPREXX CMS
I=0
RC=0
EMSGSET=ON
OPT=
Command: CMS COMMAND REXX EXEC D1 rexx CMS
Y=Y'all
COMMAND=x='Hi!'; y='Y'all';Call DumpREXX
X=Hi!
Command: CMS COMMAND EXECUTE XEDIT * EXECUTE XEDIT
$A=A
$D=D
$G=G
$L=L
$M=M
$N=N
$P=P
$T=T
$V=V
$X=X
ASTER=*
ASTRING=
ASTRING2=
C=
CMD=x DUMPREXX EXEC L1
CMD_CHAR1=X
...
```

Acknowledgements

- CMS Pipelines is the work of John Hartmann of IBM Denmark.
- CMS Pipelines was next maintained by Rob van der Heij
- VM development is now the maintainer of CMS Pipelines
- Much of the information has been gathered from the product documentation and presentations at SHARE by John and Melinda Varian of Princeton University. Melinda was a long time VMer and **the** Pipelines advocate in the VM community from before Pipelines was integrated into the VM product. She did an outstanding job of spearheading the acceptance of Pipelines and helped to drive the effort for continuing development and inclusion in the CMS product.

Summary

- CMS Pipelines is a high performance tool to enhance applications such as automation
- Pipelines are low overhead
- Pipelines are flexible allowing easy connection between stages
- Pipelines will require some practice to become proficient

Homework!

- You have some homework to do! Homework will be checked before the completion certificates are sent out after the end of the class.
- Your task:
 - Try something?
 - Create an EXEC with a Pipeline that:
 - Uses the CMS QUERY ACCESSED command
 - Exclude the R/O disks and directories
 - Writes the results to the console
 - When done, submit your EXEC:
 - SENDFILE *yourname* EXEC A TO HOMEWORK
- That's it!



Thank you for attending!

More questions? Send to: questions@velocitysoftware.com

Training, Education and Information

<https://velocitysoftware.com/educate/>

Presentations and Handouts

<https://velocitysoftware.com/pdfs/>

CMS files

- BFS
- BFSDirRectory
- BFSQuery
- BFSREPlace
- BFSSTATE
- BFSEXECUT
- Connect to a byte stream file in a byte file system
BFS can be either a read or write stage.
- Read the contents of an existing byte file system directory
- Get info from OpenEdition about current working directory
- Write records from primary input stream to replace a byte stream file
- Retrieve status information about byte stream files
- Read records from primary input stream and send each request to OpenEdition services

CMS Libraries

- LISTPDS
- MACLIB
- MEMBERS
- PDSDIRECT
- Read the directory of a CMS simulated PDS (e.g. MACLIB)
- Generate a macro library from COPY file members
- Extract members from a MACLIB, TXTLIB or a file of similar format
- Write directory information from a CMS simulated PDS

REXX

- REXXVARS
- STEM
- VAR
- VARLOAD
- Retrieve information about REXX variables
- Get or set REXX variables with the specified stem
- Get or set a REXX variable
- Set a REXX variable

Terminal

- CONSOLE
- FULLSCR
- MESSAGE
- 3270BFRA
- 3270ENC
- Read or write the terminal in line mode
- Full-screen 3270 write and read to the console or attached 3270 device
- Issue a message from a Pipeline
- Convert a 2-byte unsigned integer to the 12-bit buffer address required for some 3270 devices, or vice versa
- Prepares a 64-character translate table used to convert binary values to displayable 1-byte graphic characters

Unit Record

- PRINTMC
- PUNCH
- READER
- URO
- Print lines with machine carriage control characters
- Punch cards
- Read from a virtual card reader
- Write records to a virtual printer or virtual punch

XEDIT

- XEDIT
- XMSG
- Read or write an XEDIT in-storage file
- Issue XEDIT messages

Note: Use XEDIT stage in XEDIT macros. It is much more efficient and better performing than STACK.

Other Device Drivers

- BEGOUTput
 - Use in REXX extensions to indicate anything directed to the subcommand environment is written to the selected output stream
- DELAY
 - Suspend stream until a specific time or interval has passed
- EMSG
 - Issue messages
- HOLE
 - Discard records
- IMMCMD
 - Write the argument string from immediate commands
- ISPF
 - Access ISPF tables
- LITERAL
 - Write argument string to the output stream before copying input stream to output stream
- MDISKBLK
 - Read blocks of data from an accessed CMS minidisk
- OUTPUT
 - Write a record to the currently selected output stream
- PEEKTO
 - Inspect the next record in the selected input stream without consuming the record
- READTO
 - Read the next record from the currently selected input stream

Other Device Drivers

- QSAM
 - SETRC
 - SQL
 - SQLCODES
 - STACK
 - STORAGE
 - STRLITERAL
 - TAPE
 - VMC
 - XAB
 - X RANGE
- Read or write sequential datasets through a DCB
 - Set return code for an output subcommand
 - Interface to SQL
 - Write the last 11 SQL codes received
 - Read or write the program stack
 - Read or write virtual machine storage
 - Write argument string to the output stream before copying input stream to output stream
 - Read or write tapes at the current position
 - Write VMCF reply
 - Read or write External Attribute Buffers (XABs)
 - Create on record containing a specified range of characters

Note: Avoid STACK if possible.

Assembler Files

- ASMCONT
- ASMXPND
- Join multiline Assembler statements
- Split Assembler statements

Buffering

- BUFFER
 - Collect all records in a stage before passing any
 - Note:** Avoid BUFFER if possible - it slows down execution.
- COMBINE
- COPY
- ELASTIC
- INSTORE
- OUTSTORE
- SORT
- Combine records into one record using a logical operator
- Delay by one record the passing of records from the input stream to the output stream to prevent a pipeline stall
- Put a sufficient number of input records into a buffer to prevent a pipeline stall
- Reads records from its input stream into storage and writes a single record containing only the pointers to the records in storage.
- Unload a file loaded into storage by INSTORE
- Arrange records in ascending or descending order

“Cut and Paste”

- CHOP
- JOIN
- PAD
- SPLIT
- STRIP
- JOINCONT
- Selectively truncate records
- Concatenate groups of records
- Extend records to a specified length
- Split records into multiple records
- Remove leading or trailing characters from records
- Join records marked with a continuation string

Format Conversion

- BLOCK
- BUILDSCR
- DEBLOCK
- FBLOCK
- IEBCOPY
- PACK
- UNPACK
- Block records
- Build a 3270 data stream
- Deblock external data formats
- Block data, spanning input records
- Process an MVS unloaded data set
- Compact data
- Unpack a previously packed stream

Printer Files

- ASATOMC
- C14TO38
- MCTOASA
- OPTCDJ
- OVERSTR
- XPNDHI
- Convert ASA carriage control to machine carriage control
- Replace a set of overstruck characters with a single character
- Convert machine carriage control to ASA carriage control
- Generate a Table Reference Character (TRC) byte
- Process overstruck lines
- Highlight spaces between words

Record Format

- APLDECod
- APLENCod
- CHANGE
- ESCAPE
- REVERSE
- SCM
- SPEC
- SPILL
- SNAKE
- UNTAB
- XLATE
- Translate characters as done when read from a 3270 display
- Translate characters as done when written to a 3270 display
- Substitute one string for another
- Insert escape characters in records
- Reverse the contents of records on a character-by-character
- Line up comments and completes unclosed comments in REXX or C programs
- Rearrange the contents of records
- Split long lines at word boundaries
- Build a multicolumn page layout
- Expand tab characters (X'05') to blanks for lining up data into columns
- Translate characters based on a specified translation table

Selection Filters

- ALL
- ASMFIND
- ASMNFINF
- BETWEEN
- CASEI
- DROP
- FIND
- FRLABEL
- FRTARGET
- GETRANGE
- INSIDE
- Select records containing a specified string or specified strings defined by an expression comprising of character strings and logical operators
- Select Assembler statements that begin with a specified text
- Select Assembler statements that do not begin with a specified text
- Selects records between two specified targets including the records containing the target
- Selects records relative to a target character string regardless of case
- Discards one or more records
- Select records that begin with a specified text
- Select records that follow a specified target including the target record
- Select records starting with the first one selected by a specified stage
- Use in REXX program to extract part of a record to be processed
- Select records between two specified targets NOT including the records containing the target

Selection Filters (continued)

- LOCATE
 - Select records that contain a specified string of characters
- NFIND
 - Select records that do not begin with a specified text
- NINSIDE
 - Select records not located between two specified targets. The records containing the targets are also selected.
- NLOCATE
 - Select records that do *not* contain a specified string
- OUTSIDE
 - Select records NOT located between two specified targets. The records containing the targets are not selected.
- PICK
 - Compare a field in the primary input stream to a specified string or a second field in the record
- PREDSELECT
 - Copies a record from its primary input stream to either its primary or secondary output stream depending on the order of arrival of input records on its other input streams.
- SCANRANGE
 - Establish a token to be used by GETRANGE to parse an argument string
- SCANSTRING
 - Parse an argument string containing a delimitedString
- TAKE
 - Select one or more records from the beginning or end of the primary input stream
- TOLABEL
 - Select records that precede a specified target, not including the target record
- TOTARGET
 - Select records up to but not including the first record selected by a specified stage

Selection Filters (continued)

- UNIQUE
 - Compare the contents of adjacent records and discards or retains the duplicate records
- WHILELABEL
 - Select consecutive records that begin with a specified string. The records must be at the beginning of the input stream.
- STRASMFIND
 - Select Assembler statements that begin with a specified string of characters
- STRASMNFINFIND
 - Select Assembler statements that do NOT begin with a specified string of characters
- STRFIND
 - Select records that begin with a specified string of characters
- STRFRLABEL
 - Select records that follow a specified target including the target record
- STRNFIND
 - Select records that do not begin with a specified string of characters
- STRTOLABEL
 - Select records that precede a specified target, not including the target record
- STRWHILELABEL
 - Select consecutive records that begin with a specified string. The records must be at the beginning of the input stream.
- ZONE
 - Define a range of columns from which records are selected

Other Filters

- COUNT
- CRC
- DATECONVert
- DUPLICATE
- FMTFST
- TOKENIZE
- VCHAR
- Count bytes, blank-delimited character strings, records, or the length of the longest or shortest line
- Compute a checksum on the input stream
- Perform timestamp conversion and validation
- Copy each input record a specified number of times
- Formats a file status table (FST) entry
- Parse records according to a specified list of tokens
- Recode characters to a different number of bits per character

Control Stages

- ADDPIPE
- ADDSTREAM
- APPEND
- CALLPIPE
- COMMIT
- CONFIGURE
- GATE
- MAXSTREAM
- NOCOMMIT
- PIPCMD
- PIPESTOP
- PREFACE
- RESOLVE
- Add one or more Pipelines to a set of running Pipelines
- Define unconnected input stream, output stream or both for a stage
- Write primary input stream records to the primary output stream followed by records from a specified device driver
- Run a Pipeline and wait until it completes
- Increase a stage's commit level
- Change certain characteristics of Pipelines
- End portions of a pipeline
- Determine the highest numbered input or output stream
- Prevent automatic commit to level 0 on READTO, PEEKTO, etc
- Issue primary input stream records as pipeline subcommands
- Terminate stages waiting for an external event
- Write records from a specified device driver to the primary output stream followed by primary input stream records
- Determine if a stage is contained within and attached filter package

Control Stages

- REXX | REXXCMD
- RUNPIPE
- SELECT
- SEVER
- SHORT
- STAGENUM
- STREAM
- STREAMState
- Invoke a REXX program as a stage command
- Issue input stream records as pipelines
- Select input and/or output streams to be used on subsequent READTO, PEEKTO, SEVER, OUTPUT
- Disconnect from the currently selected stream
- Connect the currently selected input stream to the currently connected output stream
- Get the stage's relative position in the pipeline
- Specify a stream by name and have the stream number returned
- Determine the state of the specified stream

Gateways (pipe fittings)

- COLLATE
 - Match records from two input streams and writes matched and unmatched records to different output streams
- DEAL
 - Read records from primary input stream and write records to all connected output streams either sequentially or based on secondary input stream
- FANIN
 - Combine multiple input streams into a single stream in a specified order
- FANINANY
 - Combine multiple input streams into a single stream. FANINANY reads an input record from any input stream that has a record available.
- FANOUT
 - Copy primary input stream records to multiple output streams
- GATHER
 - Read records from all input streams and write to primary output
- JUXTAPOSE
 - Prefix records in the secondary input stream with records from the primary input stream
- LOOKUP
 - Find records in a reference

Gateways (continued)

- MERGE
 - Combine records from up to 10 input streams in ascending or descending order
- NOT
- OVERLAY
- SYNCHRONIZE
 - Reverse primary and secondary output streams of a specified stage command
- UPDATE
 - Read a record from each input stream and merges the records read into a single record
 - Read records from each of its input streams while each stream has a record available
 - Modify the primary input stream based on the contents of the secondary input stream

Host Command Processors

- CMS
- COMMAND
- CP
- LDRTBLS
- NUCEXT
- STARMONITOR
- STARMSG
- STARSYS
- SUBCOM
- UDP
- Issue CMS commands with full command resolution
- Issue CMS commands as if they were invoked using **Address 'COMMAND'** from REXX/VM
- Issue CP commands
- Run a compiled REXX or Assembler user-written stage that was loaded with CMS LOAD
- Call a nucleus extension as a stage
- Write monitor records from the CP *MONITOR system service
- Write lines from a CP message service
- Write lines from and sends replies to a CP system service
- Pass specified commands to a specified subcommand environment
- Allow access to a TCP/IP port

TCP/IP

- HOSTBYADdr
- HOSTBYName
- HOSTID
- HOSTNAME
- IP2SOCKA
- SOCKA2IP
- TCPCLIENT
- TCPDATA
- TCPLISTEN
- Return DNS names for IP addresses in stream
- Return IP addresses for host names in stream
- Return IP address for TCP/IP stack
- Return host name for TCP/IP stack
- Convert human-readable port number and IP address to a special sixteen-byte hexadecimal record for UDP
- Convert sixteen-byte hexadecimal record to human-readable port number and IP address
- Connect to a TCP/IP server
- Read from and write to a TCP/IP socket
- Listen on a TCP port

Assembler

- PIPCMDM
- PIPCOMMT
- PIPDESC
- PIPEPVR
- PIPGFMOD
- PIPGFTXT
- PIPINPUT
- PIPLOCAT
- PIPOUTP
- PIPSEL
- PIPSEVER
- PIPSHORT
- PIPSTRNO
- PIPSTRST

- These are a group of Assembler macros designed to make it easier to write Pipeline extensions in Assembler. I will leave researching the specifics up to you.

There is a good article on the Pipelines web site at:
<http://pucc.princeton.edu/~pipeline/sh842610.3820pack>

Service Programs

- ?
 - Display a message that describes how to obtain CMS HELP information for CMS Pipelines
- AHELP
 - View the author's online help
- FULLSCRQ
 - Query 3270 device characteristics
- FULLSCRS
 - Format output from FULLSCRQ
- HELP
 - Obtain online information about CMS Pipelines stage commands, pipeline subcommands, CMS messages, and SQL/DS return codes and commands
- QUERY
 - Display one of the following
 - current version of CMS Pipelines
 - message level
 - list of messages that have been issued
 - current level of CMS Pipelines
- TIMESTAMP
 - Determine when a record was read